
===================Histogram_1_RND(31)====================================

A HISTOGRAM CAN SHOW SOME IMPORTANT DETAILS ABOUT THE RND() FUNCTION.

====================Want_100_time_steps=======
random levels       0-> 31
Numb rnd waveforms  1
====================Build_Arrays=======
Number  Bins        0-> 31
====================Build_Noise_Signal=======
plot                nois1 vs Indx
====================Histogram_Data=======
plot                binsV vs binsN
====================Find_Ave_Rms=======
Average level       15.44
RMS level           9.05243
Average expected    15.5
RMS expected        8.94893
====================Done=======

Histogram_1_RND(31)
V1 1 0 0 dc
.control
set                       pensize              = 2
echo "====================Want_100_time_steps======="
let n                     = 100
let Nlev                  = 31
let Nrnd                  = 1
let Nbins                 = Nlev*Nrnd
echo                      "random levels       0-> $&Nlev"
echo                      "Numb rnd waveforms  $&Nrnd"
echo "====================Build_Arrays======="
unlet binsV              
unlet binsN              
unlet Indx              
unlet nois1              
let binsV                 = vector($&Nbins)*0
let binsN                 = vector($&Nbins)
let Indx                  = vector(n)
let nois1                 = vector(n)
echo                      "Number  Bins        0-> $&Nbins"
echo "====================Build_Noise_Signal======="
let index                 = 0
repeat $&n
let nois1[index]          = rnd($&Nlev) 
let index                 = index + 1
end
plot                      nois1 vs Indx
echo                      "plot                nois1 vs Indx"
echo "====================Histogram_Data======="
let index                 = 0
let hist                  = 0
repeat $&n
let indexb                = 0
repeat $&Nbins
let hist                  =  nois1[index] 
*echo  $&hist $&indexb
if (hist < indexb +.1     &  hist > indexb -.1)
let binsV[indexb]         =  binsV[indexb]  + 1
endif
let indexb                = indexb + 1
end
let index                 = index + 1
end
plot                      binsV vs binsN
echo                      "plot                binsV vs binsN"
echo "====================Find_Ave_Rms======="
let mathAve               = (Nlev)/2
let mathRMS               = mathAve/sqrt(3)
let averVal               = mean(nois1)
let noisAC                = nois1 - averVal
plot                      nois1 noisAC vs Indx
let RmsVal                = sqrt(mean(noisAC* noisAC))
echo                      "Average level       $&averVal"
echo                      "RMS level           $&RmsVal"
echo                      "Average expected    $&mathAve"
echo                      "RMS expected        $&mathRMS"
echo "====================Done======="



.endc

.end


